# QMAKE Configuration file
# Copyright (c) 2007-2010, WOSH - Wide Open Smart Home 
# by Alessandro Polo - OpenSmartHome.com
# All rights reserved.
#######################################################
# Revision $Id: woshcesrv.pro 2260 2010-04-02 15:17:29Z alex $
#######################################################

message( "WOSH CE-Server for Windows CE/Mobile 6.0" )

#####################################################
# generic configuration options
# woshcesrv gui is CE-native (not QT based)
#
TEMPLATE = app
TARGET = woshcesrv


QMAKE_CXXFLAGS_DEBUG += /wd4100


#####################################################
# set WINCE flag for sub-project files and bundles
# this is a workaround to force definition of _OS_WINCE
# and not win32, refer to common.pri@50
CONFIG += winCE

#####################################################
# set flag to include the WOSH Communication framework
# (classes like AudioMessage, TextMessage)
#
CONFIG += WOSH_COMMUNICATION
CONFIG += WOSH_SECURITY

#####################################################
# include core and interfaces, see
# framework/framework.pro for options, many
# namespaces are disabled by default
#
!include( common.pri ) {
	error( "FATAL: Project file 'common.pri' is missing! (misaligned distribution?)" )
}
# override destination folder
DESTDIR = ../bin/woshcesrv

#####################################################
# include WOSH Framework, framework options must be
# configured before following command.
#
!include( framework/framework.pro ) {
	error( "FATAL: Project file 'framework/framework.pro' is missing! (misaligned distribution?)" )
}

#####################################################
# force inclusion of DiscoveryUdp service
#
!include( bundles/services/DiscoveryUdp/DiscoveryUdp.prf ) {
	message( "*****  No bundles/services/DiscoveryUdp/DiscoveryUdp.prf file found. SKIPPED. *****" )
} 

#####################################################
# force inclusion of Windows-Mobile device
#
!include( bundles/devices/WindowsMobile/WindowsMobile.prf ) {
	message( "*****  No bundles/devices/WindowsMobile/WindowsMobile.prf file found. SKIPPED. *****" )
} 

#####################################################
# include application's core files
#
SOURCES += apps/woshcesrv/woshcesrv.cpp
HEADERS += apps/woshcesrv/wincesrv_form.h
SOURCES += apps/woshcesrv/wincesrv_form.cpp
SOURCES += apps/woshcesrv/wincesrv_form.rc

#####################################################
# additional notes for beginners
#
message( "WARNING! You may need to clear Value of Project->Properties->Linker->Advanced->Entry Point" )
